

Excel 2016: Large Data 3 vLookups



Excel 2016: Large Data 3 - vLookups

1.5 hours

In this advanced math workshop, we will work with multipart functions such as IF() and VLOOKUP() statements. We will also use Data Validation to create drop-down lists that help with data entry. We'll use the results of the list in our functions.

Class Exercises	1
vLookup	1
Title Lookup.....	2
Choose Employee.....	2
Fill in a Blank	3
Invoice	4
Data Validation: Insert or delete a drop-down list.....	8
VLOOKUP Worksheet Function	9
Description	9
Remarks.....	9
Syntax: VLOOKUP().....	9
IF Worksheet Function	11
Syntax: IF().....	11
Remarks.....	11
Logic Tree	11
Other Logic Functions.....	12
IS functions	13
Description	13
Syntax: IS().....	13
Remarks.....	13



Pandora Rose Cowart
Education/Training Specialist
UF Health IT Training

C3-013 Communicore
PO Box 100152
Gainesville, FL 32610-0152

(352) 273-5051
prcowart@ufl.edu
<http://training.health.ufl.edu>

Class Exercises

vLookup

	A	B	C	D
1	Original Data			
2	Name	Employee ID	Title	Phone #
3	Scrooge McDuck	2134-6113	Captain	(352) 555-2060
4	Donald Duck	3291-5756	First Mate	(352) 555-6108
5	Daisy Duck	9949-3960	Quartermaster	(352) 555-6615
6	Gunner 1	6670 2	Gunner 3	(352) 555-387 4
7	Huey Duck	5602-6973	Cabin Boy	(352) 555-5025
8	Louie Duck	7362-1089	Cabin Boy	(352) 555-8546
9	Dewey Duck	1910-3921	Cabin Boy	(352) 555-6756

VLOOKUP(**lookup_value**, **table_array**, **col_index_num**, **[range_lookup]**)

VLOOKUP(**Find this value**, **in this dataset**, **return data from column #**, **FALSE** (exact match))

	F	G
1	Phone Lookup	
2	Name	Phone #
3	Daisy Duck	=VLOOKUP(F3, A3:D9, 4, FALSE)
4	Dewey Duck	

Lock the Range

Cell addresses in an equation are relative to their location. When we use the fill handle or copy and paste feature the addresses move to the new location. When we fill the equation above to the next row, the formula will become:

=VLOOKUP(F4, A4:D10, 4, FALSE)

We want the first value to change, so we're now looking for "Dewey Duck", but we need the data range to stay the same. The two options we learn in the Basic 2: Math class are locking the cell addresses and naming the range.

To **Lock** a range you can type in the dollar signs (little handcuffs), or you can press function key **F4** as soon as you select the range and Excel will add the dollar signs for you. **F4**- Force!

=VLOOKUP(F3, **\$A\$3:\$D\$9**, 4, FALSE)

You need to **Name** a range before you start your equation. Select the data range, click in the name box and type the name you want for that dataset and press **Enter** on the keyboard. I used the name **Data**.

=VLOOKUP(F3, **Data**, 4, FALSE)

Title Lookup



	A	B
1	Title Lookup	
2	Name	Title
3	Daisy Duck	=VLOOKUP(A3, Data, 3, FALSE)
4	Dewey Duck	Cabin Boy
5	Donald Duck	First Mate
6	Gus Goose	Gunner

Use function key **F3** to open the name box while you're building an equation. **F3**- Find Me!

Choose Employee

	A	B
1	Choose Employee	
2	Employee:	Daisy Duck
3	Title:	=VLOOKUP(B2, Data, 3, FALSE)
4		
5	Active Employee?	

We can build a list within a cell using the **Data Validation** tool on the **Data** tab. Change the Allow option to List, and then type in the values, or the named range, or select a single column of the values you want to appear on the list.




Fill in a Blank

	A	B	C
1	Fill in a Blank		
2		Employee Name	Employee Title
3	#1:	Daisy Duck	=VLOOKUP(A3, Data, 3, FALSE)
4	#2:		#N/A
5	#3:	Louie Duck	Cabin Boy

Since there is no Employee 2 listed, we get the #N/A message saying the vLookup can't find that value.

There are three choices to deal with this.

- 1) Ignore it. I often do, I know what it means. I can use it to filter all the unmatched data.
- 2) Use a conditional formatting to make the text appear invisible by turning the font color white.



- 3) Use a nested formula with IF and ISNA.

IF(logical_test,value_if_true,value_if_false)

IF(Question, What to do if True, What to do if False)

ISNA(Value) = returns a TRUE if it's an #N/A and a FALSE if it's not an #N/A

Question: Does our vLookup return an N/A

If TRUE: If it's true, let's put a blank cell ""

If FALSE: show me what the vLookup returned

=IF(ISNA(VLOOKUP(A3, Data, 3, FALSE)), "", VLOOKUP(A3, Data, 3, FALSE))


Diagram illustrating the components of the formula:

- Question:** ISNA(VLOOKUP(A3, Data, 3, FALSE))
- If TRUE:** ""
- If FALSE:** VLOOKUP(A3, Data, 3, FALSE)

Invoice

Step 1: Name the list of Names

- Sheet "Shipping Addresses"
- Select Column A
- In the Name box, type **NameList**, press enter




The screenshot shows the Excel Name Box at the top left containing "NameList" and the formula bar containing "A". The grid below shows column A selected, with rows 1 through 4 containing names and addresses.

	A	
1	Annie Adams	6831 NW 4th
2	April Appleton	PO Box 456
3	Arnold Arlington	234 SE 45th
4	Bobbie Brown	234 Peter Pa

Step 2: Name the Addresses Range

- Sheet "Shipping Addresses"
- Select Columns A:D
- In the Name box, type **Addresses**, press enter




The screenshot shows the Excel Name Box at the top left containing "Addresses" and the formula bar containing "Annie Adams". The grid below shows columns A through D selected, with rows 1 through 4 containing names and addresses.

	A	B	C	D	E
1	Annie Adams	6831 NW 4th Ave	Gainesville, FL 32655	40620	
2	April Appleton	PO Box 456	Starke, FL 32689	41136	
3	Arnold Arlington	234 SE 45th Road	Gainesville, FL 32597	39880	
4	Bobbie Brown	234 Peter Pan Terrace	Gainesville, FL 32597	39998	

Step 3: Name the List of Items

- Sheet "Sales Items"
- Select Column A
- In the Name box, type **ItemList**, press enter




The screenshot shows the Excel Name Box at the top left containing "ItemList" and the formula bar containing "Blouses,". The grid below shows column A selected, with rows 1 through 4 containing item names and prices.

	A	B	C
1	Blouses, Blue	15.95	
2	Blouses, Red	16.45	
3	Blouses, White	15.75	
4	Pants, Blue	12.82	

Step 4: Name the Items Range

- Sheet "Sales Items"
- Select Columns A:B
- In the Name box, type **Items**, press enter




The screenshot shows the Excel Name Box at the top left containing "Items" and the formula bar containing "Blouses,". The grid below shows columns A and B selected, with rows 1 through 4 containing item names and prices.

	A	B	C
1	Blouses, Blue	15.95	
2	Blouses, Red	16.45	
3	Blouses, White	15.75	
4	Pants, Blue	12.82	

Step 5: Set up Name List

- Sheet "Sales Invoice", Cell C5
- Data Tab, Data Validation
- Allow: List
- Source: =NameList (don't forget the = sign)



Step 6: Set up Address Lookups


- Sheet "Sales Invoice"

	VLOOKUP ()	C6	C7	C8
Find	Name from cell C5	C5	C5	C5
Look in	Range "Addresses"	Addresses	Addresses	Addresses
Return	column 2, 3, 4	2	3	4
Find closest #	No, find exact	False	False	False

- C6: =VLOOKUP(C5, Addresses,2, FALSE)
- C7: =VLOOKUP(C5, Addresses,3, FALSE)
- C8: =VLOOKUP(C5, Addresses,4, FALSE)

Step 7: Set up Item List

- Sheet "Sales Invoice", Cell B11
- Data Tab, Data Validation
- Allow: List
- Source: =ItemList (don't forget the =)



- Copy/Fill formula down through Row 17

Step 8: Set up Price Lookups


- Sheet "Sales Invoice"

	VLOOKUP()	C11
Find	Item from cell B11	B11
Look in	Range "Items"	Items
Return	column 2	2
Find closest #	No, find exact	False

- C11: =VLOOKUP(B11, Items,2, FALSE)

Item	Price	Qty	Subtotal
	#N/A		

Step 9: Change equation to hide #N/A



Logical Test	Is the vLookup #N/A?	ISNA(VLOOKUP(B11,Items,2,FALSE))
If True	Leave blank	""
If False	Do the vLookup	VLOOKUP(B11,Items,2,FALSE)

- C11: =IF(ISNA(VLOOKUP(B11, Items,2, FALSE)), "", VLOOKUP(B11, Items,2, FALSE))
- Copy/Fill equation down to C17


Step 10: Set Subtotal equation

- Sheet "Sales Invoice"
- E11: =C11*D11

Item	Price	Qty	Subtotal
			#VALUE!

Step 11: Change equation to hide

- Change equation to account for blanks




Logical Test	Is the Item blank?	C11=""
If True	Leave blank	""
If False	Calculate SubTotal	C11*D11

- E11: =IF(C11="", "", C11*D11)

Data Validation: Insert or delete a drop-down list

From Office Help

To make data entry easier in Excel, or to limit entries to certain items that you define, you can create a drop-down list of valid entries that is compiled from cells elsewhere in the workbook. When you create a drop-down list for a cell, it displays an arrow in that cell. To enter information in that cell, click the arrow, and then click the entry that you want.



To create a drop-down list from a range of cells, use the **Data Validation** command in the **Data Tools** group on the **Data** tab.


1. To create a list of valid entries for the drop-down list, type the entries in a single column or row without blank cells.

For example:

	A
1	Sales
2	Finance
3	R&D

NOTE: You may want to sort the data in the order that you want it to appear in the drop-down list.

2. If you want to use another worksheet, type the list on that worksheet, and then define a name for the list.
3. Select the cell where you want the drop-down list.
4. On the **Data** tab, in the **Data Tools** group, click **Data Validation**.



5. In the **Data Validation** dialog box, click the **Settings** tab.
6. In the **Allow** box, click **List**.
7. To specify the location of the list of valid entries, do one of the following:
 - If the list is in the current worksheet, enter a reference to your list in the **Source** box.
 - If the list is on a different worksheet, enter the name that you defined for your list in the **Source** box.

In both cases, make sure that the reference or name is preceded with an equal sign (=). For example, enter **=ValidDepts**.

8. Make sure that the **In-cell dropdown** check box is selected.
9. To specify whether the cell can be left blank, select or clear the **Ignore blank** check box.

VLOOKUP Worksheet Function

From Office Help

Description

You can use the **VLOOKUP** function to search the first column of a range of cells, and then return a value from any cell on the same row of the range. For example, suppose that you have a list of employees contained in the range A2:C10. The employees' ID numbers are stored in the first column of the range, as shown in the following illustration.

	A	B	C
1	Employee ID	Department	Full Name
2	35	Sales	Yossi Banai
3	36	Production	Nicole Bousseau
4	37	Sales	Aik Chen
5	38	Operations	Axel Delgado
6	39	Sales	Suroor Fatima
7	40	Production	Gerhard Goeschl
8	41	Sales	Andreas Hauser
9	42	Operations	Nattorn Jayanama
10	43	Production	Jim Kim

If you know the employee's ID number, you can use the **VLOOKUP** function to return either the department or the name of that employee. To obtain the name of employee number 38, you can use the formula **=VLOOKUP(38, A2:C10, 3, FALSE)**. This formula searches for the value 38 in the first column of the range A2:C10, and then returns the value that is contained in the third column of the range and on the same row as the lookup value ("Axel Delgado").

The V in **VLOOKUP** stands for vertical. Use **VLOOKUP** instead of **HLOOKUP** when your comparison values are located in a column to the left of the data that you want to find.

Remarks

- When searching text values in the first column of **table_array**, ensure that the data in the first column of **table_array** does not contain leading spaces, trailing spaces, inconsistent use of straight (' or ") and curly (' or “) quotation marks, or nonprinting characters. In these cases, **VLOOKUP** might return an incorrect or unexpected value. You may be able to use the **CLEAN** and/or **TRIM** function to reformat your data.
- When searching number or date values, ensure that the data in the first column of **table_array** is not stored as text values. In this case, **VLOOKUP** might return an incorrect or unexpected value.
- If range_lookup is FALSE and **lookup_value** is text, you can use the wildcard characters — the question mark (?) and asterisk (*) — in **lookup_value**. A question mark matches any single character; an asterisk matches any sequence of characters. If you want to find an actual question mark or asterisk, type a tilde (~) preceding the character.

Syntax: VLOOKUP()

VLOOKUP(lookup_value, table_array, col_index_num, [range_lookup])

The **VLOOKUP** function syntax has the following arguments:

- **lookup_value** Required. The value to search in the first column of the table or range. The **lookup_value** argument can be a value or a reference. If the value you supply for the **lookup_value** argument is smaller than the smallest value in the first column of the **table_array** argument, **VLOOKUP** returns the #N/A error value.
- **table_array** Required. The range of cells that contains the data. You can use a reference to a range (for example, A2:D8), or a range name. The values in the first column of **table_array** are the values searched by **lookup_value**. These values can be text, numbers, or logical values. Uppercase and lowercase texts are equivalent.
- **col_index_num** Required. The column number in the **table_array** argument from which the matching value must be returned. A **col_index_num** argument of 1 returns the value in the first column in **table_array**; a **col_index_num** of 2 returns the value in the second column in **table_array**, and so on.
 - If the **col_index_num** is less than 1, **VLOOKUP** returns the #VALUE! error value.
 - If the **col_index_num** is greater than the number of columns in **table_array**, **VLOOKUP** returns the #REF! error value.
- **range_lookup** Optional. A logical value that specifies whether you want **VLOOKUP** to find an exact match or an approximate match:
 - If **range_lookup** is either TRUE or is omitted, an exact or approximate match is returned. If an exact match is not found, the next largest value that is less than **lookup_value** is returned.
 - If **range_lookup** is either TRUE or is omitted, the values in the first column of **table_array** must be placed in ascending sort order; otherwise, **VLOOKUP** might not return the correct value.
 - If **range_lookup** is FALSE, the values in the first column of **table_array** do not need to be sorted.
 - If the **range_lookup** argument is FALSE, **VLOOKUP** will find only an exact match. If there are two or more values in the first column of **table_array** that match the **lookup_value**, the first value found is used. If an exact match is not found, the error value #N/A is returned.

	VLOOKUP()	=VLOOKUP(B11,Items,2,FALSE)
Find	Item from cell B11	B11
Look in	Range "Items"	Items
Return	column 2	2
Find closest #?	No, find exact	False

IF Worksheet Function

From Office Help

Specifies a logical test to perform

Syntax: IF ()

IF(logical_test,value_if_true,value_if_false)


- *Logical_test* is any value or expression that can be evaluated to TRUE or FALSE. For example, A10=100 is a logical expression; if the value in cell A10 is equal to 100, the expression evaluates to TRUE. Otherwise, the expression evaluates to FALSE. This argument can use any comparison calculation operator.
- *Value_if_true* is the value that is returned if logical_test is TRUE. For example, if this argument is the text string "Within budget" and the logical_test argument evaluates to TRUE, then the IF function displays the text "Within budget". If logical_test is TRUE and value_if_true is blank, this argument returns 0 (zero). To display the word TRUE, use the logical value TRUE for this argument. Value_if_true can be another formula.
- *Value_if_false* is the value that is returned if logical_test is FALSE. For example, if this argument is the text string "Over budget" and the logical_test argument evaluates to FALSE, then the IF function displays the text "Over budget". If logical_test is FALSE and value_if_false is omitted, (that is, after value_if_true, there is no comma), then the logical value FALSE is returned. If logical_test is FALSE and value_if_false is blank (that is, after value_if_true, there is a comma followed by the closing parenthesis), then the value 0 (zero) is returned. Value_if_false can be another formula.

Remarks

- Up to seven IF functions can be nested as value_if_true and value_if_false arguments to construct more elaborate tests.
- When the value_if_true and value_if_false arguments are evaluated, IF returns the value returned by those statements.

=IF(A10<=100,"Within budget","Over budget")
=IF(A10=100,SUM(B5:B15),"")
=IF(B2>C2,"Over Budget","OK")
=IF(B3>C3,"Over Budget","OK")

Logic Tree



Other Logic Functions

From Office Help

TRUE

Returns the logical value TRUE.

Syntax: TRUE()

Remark: You can enter the value TRUE directly into cells and formulas without using this function.

FALSE

Returns the logical value FALSE.

Syntax: FALSE()

Remark: You can also type the word FALSE directly onto the worksheet or into the formula, and Microsoft Excel interprets it as the logical value FALSE.

AND

Returns TRUE if all its arguments are TRUE

Syntax: AND(logical1,logical2, ...)

Logical1, logical2, ... are 1 to 30 conditions you want to test that can be either TRUE or FALSE.

The arguments must evaluate to logical values such as TRUE or FALSE. If the specified range contains no logical values, returns the #VALUE! error value.

=AND(TRUE, TRUE)	TRUE
=AND(TRUE, FALSE)	FALSE

=AND(FALSE,FALSE)	FALSE
=AND(2+2=4, 2+3=5)	TRUE

OR

Returns TRUE if any argument is TRUE

Syntax: OR(logical1,logical2,...)

Logical1,logical2,... are 1 to 30 conditions you want to test that can be either TRUE or FALSE.

The arguments must evaluate to logical values such as TRUE or FALSE. If the specified range contains no logical values, returns the #VALUE! error value.

=OR(TRUE, TRUE)	TRUE
=OR(TRUE, FALSE)	TRUE

=OR(FALSE, FALSE)	FALSE
=OR(1+1=1,2+2=5)	FALSE

NOT

Reverses the value of its argument. Syntax: NOT(logical) *Logical* is a value or expression that can be evaluated to TRUE or FALSE. If logical is FALSE, NOT returns TRUE; if logical is TRUE, NOT returns FALSE.

=NOT(FALSE)	TRUE
-------------	------

=NOT(1+1=2)	FALSE
-------------	-------

IS functions

From Office Help

Description

Each of these functions, referred to collectively as the **IS** functions, checks the specified value and returns TRUE or FALSE depending on the outcome. For example, the **ISBLANK** function returns the logical value TRUE if the value argument is a reference to an empty cell; otherwise it returns FALSE.

You can use an **IS** function to get information about a value before performing a calculation or other action with it. For example, you can use the **ISERROR** function in conjunction with the **IF** function to perform a different action if an error occurs:

=IF(ISERROR(A1), "An error occurred.", A1 * 2)

This formula checks to see if an error condition exists in A1. If so, the **IF** function returns the message "An error occurred." If no error exists, the **IF** function performs the calculation A1*2.

Syntax: IS()

The **IS** function syntax has the following argument:

- **Value** -Required. The value that you want tested. The value argument can be a blank (empty cell), error, logical value, text, number, or reference value, or a name referring to any of these.

FUNCTION	RETURNS TRUE IF
ISBLANK	Value refers to an empty cell.
ISERR	Value refers to any error value except #N/A.
ISERROR	Value refers to any error value (#N/A, #VALUE!, #REF!, #DIV/0!, #NUM!, #NAME?, or #NULL!).
ISLOGICAL	Value refers to a logical value.
ISNA	Value refers to the #N/A (value not available) error value.
ISNONTEXT	Value refers to any item that is not text. (Note that this function returns TRUE if the value refers to a blank cell.)
ISNUMBER	Value refers to a number.
ISREF	Value refers to a reference.
ISTEXT	Value refers to text.

Remarks

- The value arguments of the **IS** functions are not converted. Any numeric values that are enclosed in double quotation marks are treated as text. For example, in most other functions where a number is required, the text value "19" is converted to the number 19. However, in the formula **ISNUMBER("19")**, "19" is not converted from a text value to a number value, and the **ISNUMBER** function returns FALSE.
- The **IS** functions are useful in formulas for testing the outcome of a calculation. When combined with the **IF** function, these functions provide a method for locating errors in formulas.